
Hosting .NET Web Services

Get started in using the Winsock control to host Web Services as I walk you through the sample code attached to this article. 

Download 694-CF-DEV.exe.

What You Need

· Microsoft® eMbedded Visual Tools. 

· A live Internet connection from your Pocket PC. 

· To call Web Services from your PC, you can download Pocket SOAP (Simple Object Access Protocol). 

Gotchas

· The implementation to host Web Services using SOAP in this article is very simplified and needs to be developed further to fully support any client accessing them. For example, the lack of a WSDL (Web Services Description Language) support prevents the Web Services from being called from Microsoft SOAP Toolkit 2.0 and the beta versions of Microsoft Visual Studio® .NET. 

Languages Supported

English 

Why Host Web Services?

In my previous article Transfer Data the .NET Way, I provided samples for accessing Web Services (using SOAP) from the Pocket PC. That is probably the most common way that you will use Web Services on a Pocket PC because this is a great way to make the device the point of integration. 

The next logical step would be to also host those Web Services on your Pocket PC. You can probably think of more reasons than I can (don't hesitate to send me an e-mail if you do), but here are a few of my suggestions: 

· Your personal calendar Web Service; that anyone (or probably your trusted friends) can call to check if you are available at a specific time and maybe even create an appointment with you. 

· Your personal contacts Web Service; that can be called to extract contact information. 

· Your current availability status Web Service; let's say you are a field service technician and your backoffice or even the customers want to check if you are available. In a local application, you could mark yourself as "online", "busy", "be right back", "away", "on the phone", "out to lunch", and so forth. 

Also, if you want to be able to do P2P (peer-to-peer) applications, you will need to both call and host Web Services on your Pocket PC. The number of P2P applications will increase rapidly as soon as we have this support on Pocket PCs. Some examples of P2P applications are: 

· File Sharing: exchange files with friends directly without any need for a server. If you have your friend's ID (IP address), you could query her Pocket PC for available files (that she "published" by putting them in a certain folder) and then download the file to your Pocket PC. 

· Messaging: chatting with your colleagues while on the go without any supporting server. You could even set up virtual meetings with several people without the need for a chat server. 

· Gaming: imagine if you could connect directly with other players to play your favorite game together, wirelessly. 

Even if there are products that can provide similar functionality (like Odyssey Software's ViaXML) already today, I haven't seen any that can publish Web Services using SOAP on the Pocket PC. I have therefore put together a sample that shows you how a simple Web Service (using SOAP) can be provided using Microsoft eMbedded Visual Basic®. 

Sample Web Service

When you run the sample application, it looks like this: 

Sample SOAP request and response.


And in the above figure you can see that we have made a Web Service (SOAP) call from a client on another machine (a PC) using Pocket SOAP. Even if you are not familiar with SOAP, you can see in the request that a call is made to a method called Multiply with two parameters (a and b) with the values 3 (a) and 5 (b). In the response, the result is returned (15). And whatever number you provide with the request, the two numbers will get multiplied and returned. The really cool thing is that any client (PC, Pocket PC, mainframe, etc.) that can make a SOAP request, can now call your Pocket PC from anywhere in the world (provided you are connected to the Internet). 

Code Walkthrough

The first thing you need to publish a Web Service is a Web server. You probably know that there is no Web server in a Pocket PC, and therefore we need to create one first. You can do that by using the Winsock control in eMbedded Visual Basic. If you add one to a form and name it wsoServer, you can use the following code to activate it on port 80, the port used for HTTP (Hypertext Transfer Protocol): 

Private Sub Form_Load()

  wsoServer.LocalPort = 80

  wsoServer.Listen

End Sub

And as soon as a HTTP request comes in, you can accept it by: 

Private Sub wsoServer_ConnectionRequest()

  wsoServer.Accept

End Sub

The next thing to do, is to wait for the request to arrive and when it does, we can handle it like this: 

Private Sub wsoServer_DataArrival(ByVal bytesTotal As Long)

  Dim ls As String

  Dim lsHead As String

  Dim Params As Variant

  ' Get Request

  wsoServer.GetData ls

  txtRequest.Text = ls

  ' Parse Request

  ls = Mid(ls, InStr(ls, "<"))

  Params = SOAPParse(ls)

  ' Do Web Service

  ls = WebService(Params)

  ' Set Response

  ls = SOAPResponse(ls)

  ' Send Response

  lsHead = "HTTP/1.1 200 OK" & vbCrLf

  lsHead = lsHead & "Content-Length: " & CStr(Len(ls)) & vbCrLf

  ls = lsHead & vbCrLf & ls

  wsoServer.SendData ls

  txtResponse.Text = ls

End Sub

This event is called as soon as any data is received on the specified port that we listen to (in this case 80). We start by getting the data using the GetData method on the Winsock control. We output this data in the TextBox control (see figure above). We remove the HTTP header using the Mid function and then we parse the SOAP request into a Variant array using the SOAPParse function (see below). Then we call the WebService function that implements the functionality of our Web Service. A response is created using the SOAPResponse function and it is sent back to the client with a correct HTTP header. Also, the response is output in another TextBox control (see figure above). 

The implementation of the SOAPParse function looks like this: 

Public Function SOAPParse(ByVal Request As String) As Variant

  Dim loXMLDoc As DOMDocument

  Dim loNodeList As IXMLDOMNodeList

  Dim loErrXML As IXMLDOMParseError

  Dim lnodX As IXMLDOMNode

  Dim lnodY As IXMLDOMNode

  Dim i As Integer

  Dim Params()

  ' Create XML document

  Set loXMLDoc = CreateObject("Microsoft.XMLDOM")

  ' Load request data

  loXMLDoc.loadXML Request

  ' Get Method and Namespace

  Set lnodX = loXMLDoc.documentElement.childNodes(0).childNodes(0)

  psMethod = lnodX.baseName

  psNamespace = lnodX.namespaceURI

  ' Get Parameters

  Set loNodeList = loXMLDoc.documentElement.childNodes(0).childNodes(0).childNodes   

  ReDim Params(1, loNodeList.length - 1)

  For i = 0 To loNodeList.length - 1

    Set lnodX = loNodeList(i)

    Params(0, i) = lnodX.nodeName

    Params(1, i) = lnodX.Text

  Next 'i

  ' Return parameter array

  SOAPParse = Params

End Function

And you can see that we read the XML (Extensible Markup Language) using a XML DOM (Document Object Model) and turn it into a two-dimensional array containing the two passed parameters. In the (0, x) array elements, we place the parameter name and in the (1, x) array elements, we place the parameter value. The WebService function implementation looks like this: 

Private Function WebService(ByRef Params As Variant) As String

  WebService = CStr(CInt(Params(1, 0)) * CInt(Params(1, 1)))

End Function

This is the place where the two parameters are multiplied and the result is returned. The SOAPResponse function creates the SOAP response: 

Public Function SOAPResponse(ByVal Result As String) As String

  Dim lsResponse As String

  ' Set payload

  lsResponse = _

  "<?xml version='1.0' encoding='UTF-8' standalone='no'? >" & vbCrLf & _

  "<S:Envelope S:encodingStyle='http://schemas.xmlsoap.org/soap/encoding/'" & _

  " xmlns:S='http://schemas.xmlsoap.org/soap/envelope/'>" & vbCrLf & _

  "<S:Body>" & vbCrLf & _

  "  <m:" & psMethod & "Response xmlns:m='" & _

  psNamespace & "'>" & vbCrLf & _

  "    <Result>" & Result & "</Result>" & vbCrLf & _

  "  </m:" & psMethod & "Response>" & vbCrLf & _

  "</S:Body>" & vbCrLf & _

  "</S:Envelope>"

  ' Return response

  SOAPResponse = lsResponse

End Function

And when the response is returned using the SendData method on the Winsock control, the following event gets called: 

Private Sub wsoServer_SendComplete()

  wsoServer.Close

  wsoServer.Listen

End Sub

Where we close the connection and restart the control to listen for new requests. 

Using Pocket SOAP the test client to test the Web Service that we have implemented looks like this: 

TestPocketPC

Function TestPocketPC()

  dim x, soap, t

  ' Create objects

  set soap = CreateObject("PocketSOAP.Envelope")

  set t = CreateObject("PocketSOAP.HTTPTransport")

  ' Set Envelope

  soap.methodName = "Multiply"

  soap.URI = "urn:Foo"

  ' Set parameters

  soap.CreateParameter "a", "3"

  soap.CreateParameter "b", "5"

  ' Make Call

  t.Send "http://ipaq/", soap.serialize

  x = t.Receive

  soap.parse x

  ' Show result

  wscript.echo "Result = " & soap.Parameters.Item(0).Value

End Function

An important note here is that my Pocket PC have been associated with the name "ipaq" and there are several ways to do that. If your Pocket PC is logged on to a Microsoft Windows NT®/Microsoft Windows® 2000 network, the name of your Pocket PC can be used in the URL. And if not, you can add your Pocket PC's name to the Hosts file (located in the \WINNT\system32\drivers\etc folder on your PC). 

Implement Future Solutions Now

You can be quite sure that this will be possible to do natively on Pocket PCs in the future. If you start thinking, or maybe even implementing, solutions that utilize this technology, you will be well prepared for the future. Even if the sample above is intentionally made very simple, you could extend it to create real-world Web Services solutions for Pocket PC. 

Conclusion

You now have the power to both consume (in the articles mentioned above) as well as host Web Services on your Pocket PC. This power opens up numerous possibilities, especially regarding P2P applications, and you are the one that can make them come true
